


マイコンビンゴマシンのプログラミング

●ビンゴのプログラム

マイコンビンゴマシンで動くビンゴのプログラムを作ります。




USB 端子にキーボード、ビデオ端子にビデオケーブルとモニタをつないで、電源を入ると、IchigoJam の画面が表示されます。



最初に、表示をリセットします。

```
OUT 8,0<Enter>
```

表示される数字が「80」→「00」になります。



●数字を表示する

まずはかんたんに数字を表示してみましょう。

```
OUT #12<Enter>
```

7セグメントLEDに数字が表示されます。
数字をいろいろ変えて、試してみましょう。



試してみるとわかりますが、このように表示されます

```
OUT # (10の位の数字) (1の位の数字)
```

※「#」(シャープ、ハッシュ)は16進数を表します。

10進数は、10集まると次の位へ上がります。

16進数は、16集まると次の位へ上がります。

10進数	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
16進数	0	1	2	3	4	5	6	7	8	9	A	B	C	D	E	F	10

※このビンゴマシンでは、A~Fは表示されません。

●ランダムな数字を表示する

毎回同じ数字が出てしまつてはビンゴマシンになりません。

1~75までのランダムな数字を表示してみましょう。

2行のプログラムを入力します。

```
10 N=RND(75)+1<Enter>
```

Nに1~75のランダムな数字を入れる

```
20 OUT N/10*16+N%10<Enter>
```

Nを16進数2桁に変換して
ポートへ出力

入力できたら、

```
RUN<Enter>
```

と入力するか、F5キーを押します。

プログラムが実行されて、7セグメントLEDに数字が表示されます。

実行するたびに数字が変わります。



Nは変数です。数字を入れておく「入れ物」「箱」だと思ってください。

「RND(75)」は0~74までの乱数(らんすう)を出します。毎回数字が変わります。

「RND(75)+1」とすることで1が足されて、1~75のランダムな数になります。

●ボタンを押したら数字を出す

ボタンを押したら数字を出すようにします。

```

5 IF !BTN() CONT
10 N=RND(75)+1
20 OUT N/10*16+N%10
25 WAIT 10
30 GOTO 5
    
```

ボタンが押されるまで待つ

時間待ち

5行へもどる

※青い色がついた行だけ追加で入力します。

「RUN」または F5 キーで実行します。

ビンゴマシンのボタンを押すと、くり返し数字が出ます。



パッと数字が出て止まるだけでは面白くないので、数字がパカパカと変化してから止まるようにします。

```

5 IF !BTN() CONT
9 FOR L=1 TO 10
10 N=RND(75)+1
20 OUT N/10*16+N%10
25 WAIT 10
27 NEXT
30 GOTO 5
    
```

Lを1～10まで10回くりかえす

くり返しここまで

「RUN」または F5 キーで実行します。

ボタンを押すと、数字が10回くり返して出してから止まります。

さらにボタンを押すと、くり返し数字が出ます。

数字を出す時に、効果音をつけてみましょう。

```

22 BEEP 10,2
    
```

音を出す

「RUN」または F5 キーで実行します。

ボタンを押すと、数字が出る時に「ピッ」と音が出ます。

●一度出た数字はもう出ないようにする

ビンゴマシンでは、一度出た数字はもう出ません。

一度出た数字を覚えておいて、次は出さないプログラムにします。

3 CLY:OUT 0:OUT 8,0	変数をクリア:数字をクリア
5 IF !BTN() CONT	
6 SRND TICK()	乱数を初期化
9 FOR L=1 TO 10	
10 N=RND(75)+1	
15 IF [N]=1 GOTO 10	これまで出た数字だったらやり直し
20 OUT N/10*16+N%10	
22 BEEP 10,2	
25 WAIT 10	
27 NEXT	
28 [N]=1	出た数字を記憶
30 GOTO 5	

「RUN」または F5 キーで実行します。一度出た数字はもう出なくなります。

[N] のように大かっこが付いた変数を、**配列(はいれつ)変数**といいます。「〇番目」と番号がついた数」と思えばいいです。**[0]~[101]**まで 102 個あります。

プログラムがスタートした直後は、3 行の「**CLY**」で変数が全てクリアされるので、配列変数 **[0]~[101]** も全部 0 になっています。

何か数字を出した後、27 行で「**[N]=1**」として、N 番目の配列変数を 1 にしています。

15 行でそれを判定して、もし一度出た数字だったら配列変数 **[N]** が 1 になっているので、その場合は 10 行へ戻って、ちがう数字を出し直します。

●電源 ON で自動実行する

プログラムの先頭に 1 行追加します。

1 @ARUN: ^BINGO	自動実行:プログラムのタイトル
-----------------	-----------------

入力したら、プログラムをファイル 0 番に保存します。

SAVE 0

ちゃんと保存できると、「Saved XXbyte」(XX は数字)と表示されます。

保存できたら、1 回電源スイッチを切って、もう一度電源を入れます。

ビンゴのプログラムが自動的に動きます。

★できる人は:レビューモードを追加

ビンゴをやっていると、「これまでに出了数字を確認したい」ことがあります。
そのためのレビューモードを追加してみましょう。

```

1  SARUN: ^BINGO
3  CLY:OUT 0:OUT 8,0
5  IF !BTN() CONT
6  SRND TICK():CLT 乱数をクリア:タイマーをクリア
7  IF BTN() CONT ボタンがはなされるまで待つ
8  IF TICK()>60 GSB 40:GOTO 5
9  FOR L=1 TO 10 ボタンが長押しされていたらレビューモードへ
10 N=RND(75)+1
15 IF [N]=1 GOTO 10
20 OUT N/10*16+N%10
22 BEEP 10,2
25 WAIT 10
27 NEXT
28 [N]=1
30 GOTO 5
40 N=0 レビュー数字をリセット レビューが最後まで
50 N=N+1 レビュー数字を次へ 行ったら戻る
60 IF N>75 BEEP 30,30:WAIT 60:OUT 0:RTN
70 IF ![N] GOTO 50 出していない数字ならレビューせず次へ
80 OUT N/10*16+N%10 数字をレビュー
90 BEEP 30,2 効果音を出す
100 WAIT 10 時間待ち
110 IF !BTN() CONT ボタンが押されるまで待つ
120 IF BTN() CONT ボタンがはなされるまで待つ
130 GOTO 50 次のレビュー数字へ
    
```

「RUN」または F5 キーで実行します。

ボタンを押して、何度か数字を出した後、ボタンを 1 秒以上長押ししてはなします。

レビューモードになり、ボタンを押すとこれまで出了数字が確認できます。

確認が終わるとブザーが長く鳴り、「00」表示に戻ります。またボタンを押して、次の数字を出してください。

動作が確認できたら、プログラムを保存しましょう。

SAVE 0

ちゃんと保存できると、「Saved XXbyte」(XX は数字)と表示されます。

★できる人は:100m ダッシュゲーム

まず、ビンゴのプログラムをファイル 1 番に保存しておきます。

SAVE 1

ちゃんと保存できると、「Saved XXbyte」(XX は数字)と表示されます。

今動いているビンゴのプログラムを消去します。

NEW

※ビンゴのプログラムはファイル 1 番に保存されているので、「LOAD 1」でまた呼び出せます。

以下の「100m ダッシュゲーム」を入力してみましょう。

```

10  ' *100M DASH      タイトルコメント
20  CLY:OUT 8,0      変数クリア:表示クリア
30  FOR T=3 TO 1 STEP -1
40  OUT 160+T:BEEP
50  WAIT 60
60  NEXT
70  OUT 0:CLT      表示クリア:タイマーリセット
80  FOR X=1 TO 100  1~100m までくりかえし
90  IF !BTN() CONT  ボタンが押されるまで待つ
100 IF BTN() CONT   ボタンがはなされるまで待つ
110 OUT X/10*16+X%10  走った距離を表示
120 NEXT           くりかえしここまで
130 T=TICK()      タイムを読み出し
140 OUT 0         表示クリア
150 BEEP 10,30    ゴール効果音
160 WAIT 60      時間待ち
170 S=T/60:R=T%60/6  タイムの秒数と、0.1秒の位を計算
180 OUT S/10*16+S%10:CLT  秒数を表示
190 IF BTN() RUN   ボタンが押されたらリプレイ
200 IF TICK()<60 GOTO 190 1秒たつまでボタン待ち
210 OUT R*16+10:CLT  0.1秒の位を表示
220 IF BTN() RUN   ボタンが押されたらリプレイ
230 IF TICK()<60 GOTO 220 1秒たつまでボタン待ち
240 GOTO 180      タイム表示をくりかえし
    
```

3、2、1 とダウンカウント

入力できたら、「RUN」または F5 キーで実行します。
 実行すると、「3」「2」「1」とダウンカウントした後、「00」表示になります。
 ボタンを 100 回連打して 100m 走ってください。
 ゴールすると、タイムが「秒の単位 2 桁」→「0.1 秒の単位 1 桁」で交互に表示されます。
 ボタンを押すとリプレイします。

100m ダッシュのプログラムが動いたら、ファイル 2 番に保存します。

SAVE 2

ちゃんと保存できると、「Saved XXbyte」(XX は数字)と表示されます。

★プログラムセクタ

このままだと、キーボードをつながないと、ビンゴマシンと 100m ダッシュの両方を実行できません。

ボタンでも2つのプログラムを選んで実行できるように、ファイル 0 番に**セクタ**のプログラムを入れます。

今動いている 100m ダッシュのプログラムを消去します。

NEW

以下の「セクタ」のプログラムを入力します。

```
10 @ARUN: ^*SELECTER
20 CLV:OUT 8,0:G=1
30 CLT:OUT 160+G
40 IF BTN() BEEP:WAIT 60:LRUN G
50 IF TICK() < 60 GOTO 40
60 G=G+1:IF G>3 G=1
70 GOTO 30
```

入力できたらファイル 0 番に保存します。

SAVE 0

ビンゴマシンの電源スイッチを OFF にして、もう一度 ON にします。
 数字が「1」→「2」→「3」→…と順番に表示されます。
 ボタンを押すと、表示されたファイル番号(1~3)のプログラムを実行します。
 「1」を選べばビンゴマシン、「2」を選べば 100m ダッシュが動くはずですが。
 プログラムを切り替えたい時は、電源スイッチを OFF にして、もう一度 ON にします。